

## Online Supplementary Material

### **Environmental control of the dominant phytoplankton in the Cariaco basin: a hierarchical Bayesian approach**

C.M. MUTSHINDA, L. TROCCOLI-GHINAGLIA, Z.V. FINKEL, F.E. MÜLLER-KARGER & A.J. IRWIN

#### **Description**

This file contains the OpenBUGS (Thomas et al. 2006) code for fitting the hierarchical Bayesian model (Gelman et al. 2003) described in the paper 'Environmental control of the dominant phytoplankton in the Cariaco basin: a hierarchical Bayesian approach' by Mutshinda et al.

In the BUGS code provided below, Nobs denotes the number of observations per cruise, which is 469 i.e. 67 species  $\times$  7 observations/species. The 7 observations correspond to the 7 depths on which phytoplankton abundances and environmental data are observed. Ncruise is the number of cruises (167), and  $\beta_{s,j}$  is the effect of the  $j$ th environmental variable (irradiance, temperature, silicic acid concentration, phosphate concentration, nitrate concentration and pH) on the log-abundance of species  $s$ .

The Bayesian freeware OpenBUGS (Thomas et al. 2006; Lunn et al. 2009) is freely available at URL <http://www.openbugs.info/w/>

## BUGS code for the hierarchical Bayesian model

```
#####  
model{  
  # Likelihood  
  for (i in 1:Nobs) {  
    for (j in 1:NCruise) {  
      y[i,j] ~ dnorm(mu[i,j], tau[Species[i]])  
      mu[i,j] <- alpha[Species[i]] +  
        beta[Species[i],1]*irrad[j,Depth[i]] +  
        beta[Species[i],2]*Temp[j,Depth[i]] +  
        beta[Species[i],3]*SiOH4[j,Depth[i]] +  
        beta[Species[i],4]*PO4[j,Depth[i]] +  
        beta[Species[i],5]*NO3[j,Depth[i]] +  
        beta[Species[i],6]*Salinity[j,Depth[i]] +  
        beta[Species[i],7]*pH[j,Depth[i]]  
    }  
  }  
  # Prior specification  
  for (s in 1:Nspecies) {  
    beta[s, 1:Nvar] ~ dnorm(m[Species[s], 1:Nvar], taub[,])  
    sigma2[s] ~ dgamma(a, b)  
    tau[s] <- 1/sigma2[s]  
  }  
  
  # Precision matrix of environmental effects  
  taub[1:Nvar, 1:Nvar] ~ dwish(R[,], Nvar)  
  
  # Covariance matrix of environmental effects  
  Omega[1:Nvar, 1:Nvar] <- inverse(taub[,])  
  for (s in 1:Nspecies) {  
    alpha[s] ~ dnorm(0, tau.a) # Intercept for species s  
    for (j in 1:Nvar) {  
      m[s, j] <- 0  
      # m[s, j] is the prior mean of beta[s,j]  
    }  
  }  
  for (k in 1:Nvar) {  
    for (j in 1:Nvar) {  
      rho[k,j] <- Omega[k,j]/sqrt(Omega[k,k]*Omega[j,j])  
    }  
  }  
  tau.a <- 1/sigma2_alpha  
  sigma2_alpha ~ dgamma(1, 1)  
  a ~ dgamma(1, 1)  
  b ~ dgamma(1, 1)  
}  
#####
```

## References

- Gelman A, Carlin JB, Stern HS, Rubin DB 2003. Bayesian Data Analysis, 2nd Edn. London: Chapman & Hall, p 117–56.
- Gilks WR, Richardson S, Spiegelhalter DJ. 1996. Markov Chain Monte Carlo in Practice. London: Chapman and Hall, p 1–19.
- Lunn D, Spiegelhalter D, Thomas A, Best N. 2009. The BUGS project: Evolution, critique and future directions (with discussion). *Statistics in Medicine* 28:3049–82.